Paper 214-28

POSTING PROJECT STATUS TO WEB THROUGH SAS® PROGRAMMING

David Shen, DZS Consulting, Inc.
Zaizai Lu, AstraZeneca LP

ABSTRACT

In clinical trial studies, it is important to learn how the project is going on. Information such as how many programs have been developed, how many data listings and tables have been generated, and how much validation is finished, is very helpful to adjust resources and check the status of a project. A small compact program has been developed to meet such a need. This program, with a specified directory provided, can search all files from the current directory and all its subdirectories for information such as filename, path, date and time created. User can select files by file type, time period. Finally, the program will generate an HTML file that contains file list and summary information. A hyper-linkage makes it possible to open and view the file of interest on the web site. This HTML file posted on Web can dynamically report the progress status of the project.

INTRODUCTION

The status of the project is very important to adjust resources and make a business decision. However, it is not static. It keeps changing with the generation of files such as patient enrollment status, CRF completion status, client requirement documents, design, test script, program files, output and log documents. The dynamic web-based file summary is capable of delivering the information timely to satisfy the need for learning the project status. The information about the project can be refreshed and updated continuously with newly-generated files as the project goes on.

BASIC PROGRAM

The program can be divided into three parts:

- Collect file information
- Filter and organize the information
- Post them on web.

```
%let projloc   =; ❶
%let strtdate  =;
%let stopdate  =;
%let filetype  =;
%let sortby    =;
%let htmlfile  =;

option noxwait xsync;

proc datasets lib=work kill;
quit;

filename filepath "&htmlfile" ;


%sysexec  "dir /s &projloc">c:\temp.dat; ❷
data _readin;
  infile 'c:\temp.dat' length=len;
  input @;
  input line $varying200.len;

  if index(line, 'Directory of') then do; ❸
PATH=scan(upcase(line), 3, ' ');
     retain path;
     delete;
  end;
   if line ne ' ' and indexc(line, '.') and
      not index(line, '<DIR>');
run;

data _readed;
  format DATE mmddyy10. TIME time.; ❹
  set _readin;
  date=input(scan(line, 1, ' '), mmddyy10.);

  if indexc(scan(line, 2, ' '), 'a') then do;
      time=input(compress(scan(line, 2, ' ' ),
                 'a'), time.);
      if time>='12:00't then time=time-'12:00't;
  end;
  if indexc(scan(line, 2, ' '), 'p') then do;
      time=input(compress(scan(line, 2, ' ' ),
                 'p'), time.);
      if time<'12:00't then time=time+'12:00't;
  end;

  FIlE=line;
  do i=1 to 3;  ❺
     sublen=length(scan(left(file), 1, ' '));
     file=substr(left(file), sublen+1);
  end;

  if "&filetype"^=' ' then do;
    if upcase(scan(file, -1, '.')) =
          "%upcase (&filetype)";
  end;
  drop i sublen line;
run;

data _subset;  ❻
  set _readed;
  if "&strtdate"^=" " then
       lowdate=input("&strtdate", mmddyy10.);
  else lowdate=.;

  if "&stopdate"^=" " then
       highdate=input("&stopdate", mmddyy10.);
  else highdate=today();

  if lowdate<=date<=highdate;
run;

data _list;
  set _subset;
  if path^=' ';
  TYPE=upcase(scan(file, -1, '.'));
```

```
  FILE='<ahref="'||trim(left(path))||'\'
       ||trim(left(file))||'">'||trim(left(file))
       ||'</a>';  ❼
run;

proc sort data=_list;
by &sortby;
run;

proc print; run;

ods listing close;
ods html file  = filepath
         style = Sasweb; ❽
  proc print data=_list obs='OBS';
    title "LIST OF %upcase(&filetype) FILES
           AT %upcase(&_loc)";
    var  date time file path ;
    by   &sortby;
  run;

  proc sql ;  ❾
    title "SUMMARY OF %upcase(&filetype) FILES
           AT %upcase(&_loc)";
    select path, type, count (file) as TotalFiles
         from _list
            group by path, type
              order by path, type, totalfiles;
  quit;

ods html close;
ods listing;
```

❶Macro parameters, which allow users to select project directory, file type and data period. If file type is not defined, then all the files will be listed. Sortby allows users to show the file information in different orders such as path, file type or date and time file created.

❷Usually there are three ways to be used to gather file information from the system: pipe, output redirection and SAS function. Pipe does not work for Window 98. Redirection works for Windows 98, NT, 2000. But the output may be arranged differently according to the operating system. Attention should be paid when extracting data from the redirection output. SAS function is portable, however, it just tells the information in the current directory, so a loop is needed to collect information from its subdirectory, or sub-subdirectory, etc. The output redirection method is used here by %sysexec. The statement %sysexec may be replaced by function call system() which is used in data *null* step. The output from DOS command DIR is redirected into a data file temp.dat. Its contents are as below:

```
Volume in drive C has no label.
Volume Serial Number is E80B-BACF

 Directory of C:\WKU\adhoc

09/13/2002  09:32a      <DIR>          .
09/13/2002  09:32a      <DIR>          ..
08/23/2002  05:31p           1,017,344 BLOOD CHEM.doc
08/23/2002  05:32p             272,384 LIPOPROTEIN.doc
08/23/2002  11:13a              44,544 LISTING OF
DEMOGRAPH.doc
               3 File(s)      1,334,272 bytes

Directory of C:\WKU\sas\program

09/13/2002  09:51a      <DIR>          .
09/13/2002  09:51a      <DIR>          ..
08/06/2002  10:42a               2,754 adversevent.sas
08/21/2002  02:02p               3,642 bloodchem.sas
07/03/2002  02:21p               3,507 discontinue.sas
08/12/2002  11:36a               3,318 vital.sas
08/14/2002  04:24p               2,325 vitals.sas
               5 File(s)         15,546 bytes
```

❸Try to read the data file into a SAS dataset. Since the lengths of the records in data file are different, informat varying is applied here. The function index () is used to find the path and files.

❹Since DOS time format is different from SAS time format, so a conversion is conducted. For example, convert 02:02p to 14:02.

❺The DO loop is used to extract file name. Since file name may contain space, for example 'Blood Chem.Doc', the simple scan function does not work in this case. The result of scan(line, 3, ' ') will only be 'Blood', not the whole file name. The files will be filtered if file type is specified. For example, if filetype=SAS, then only SAS programs will be collected and analyzed. By default, all the file information will be gathered.

❻If the user wants to know how many files generated in a specified period, then start and stop date can be defined for this purpose. By default, there is no time period defined.

❼Build a hyper-link between the file name and file location.

❽Use ODS html to make the html file. SAS software provides several defined styles such as brick, brown, d3d, etc. They can be listed to output Window by

```
proc template;
  list styles;
run;
```

Users may select one of those styles based on their preferences.

❾PROC SQL is used to summarize the file information.

EXAMPLE

If we run the program by setting:

```
%let projloc  = C:\WKU
%let strtdate  = 8/20/02;
%let stopdate =;
```

```
%let filetype  =;
%let sortby    =path;
```

The result is shown at the end of this paper.

CONCLUSION

SAS ODS HTML makes it very easy to display the status of the project on the web site dynamically. The hyper-link created in HTML enables users to open and view files on site conveniently. The instantly available project status will be very helpful to make a business decision.

CONTACT INFORMATION

Zaizai Lu
903 Parkview Dr.
King of Prussia, PA 19406
e-mail: zz_lu@hotmail.com



SAS Output - Microsoft Internet Explorer - [Working Offline]

File Edit View Favorites Tools Help

Back Search Favorites History

Address C:\temp\A\list1.html Go Links

LIST OF FILES AT C:\WKU

PATH=C:\WKU\ADHOC

OBS	DATE	TIME	FILE	PATH
1	08/23/2002	17:31:00	BLOOD CHEMISTRY.doc	C:\WKU\ADHOC
2	08/23/2002	17:32:00	LIPOPROTEIN.doc	C:\WKU\ADHOC
3	08/23/2002	11:13:00	LISTING OF DEMOGRAPH.doc	C:\WKU\ADHOC

PATH=C:\WKU\DOCS\COMM

OBS	DATE	TIME	FILE	PATH
4	08/26/2002	8:47:00	listings.xls	C:\WKU\DOCS\COMM
5	08/29/2002	12:39:00	tables.xls	C:\WKU\DOCS\COMM
6	08/26/2002	17:50:00	Table of Contents.doc	C:\WKU\DOCS\COMM

SAS Output - Microsoft Internet Explorer - [Working Offline]

File Edit View Favorites Tools Help

Back Search Favorites History

Address C:\temp\A\list1.html Go Links

SUMMARY OF FILES AT C:\WKU

PATH	TYPE	TotalFiles
C:\WKU\ADHOC	DOC	3
C:\WKU\DOCS\COMM	DOC	1
C:\WKU\DOCS\COMM	XLS	2
C:\WKU\DOCS\PROCS	DOC	1
C:\WKU\GRAPH	PDF	1
C:\WKU\LISTLOG	LOG	2
C:\WKU\LISTLOG	LST	2
C:\WKU\SAS\FORMATS	SAS	1
C:\WKU\TRANSFER	ZIP	3